
** Table A4: Separation
********************************************************************************
* Regression Analysis
********************************************************************************


* Separation results

eststo clear
	
	
use "$OUTDATA/NFHS_sample_withdivorce.dta", clear 


* Macros

* Outcomes

global dec " w_dec n_w_dec sayl hcdec huss saysc "
global viol " injuries n_inj viol_p2 viol_p1 viol_s viol_e "

global marriage "age_marr age_gap educ_gap height_diff workh h_goodjob h_earnless drunk_often"
global wealth "asset1 asset2 modern_cook modern_roof agrland_hec wealth_index sl_index "
global hc "educw primary height lowheight"
global reporting "viol_byothers viol_natal told_anyone sought_help1 sought_help2 sought_help3"

* Covariates and FE
global cov "rural christian hindu scstbc"
global cov1 "wealth_index hhsize educw"
global cov2 "ageh educh workw workh"
global fe1   "i.state i.yy_1marr"
global fe2   "i.state i.yearb"
global did "i.dow_cohort##i.hindu1"

global cluster "state"

global graph "ylabel(, nogrid) scheme(s2mono) plotregion(fcolor(white) lcolor(white) margin(small)) graphregion(fcolor(white) lcolor(white) ifcolor(white) ilcolor(white))"


* Labels for Tables

label var w_dec "\shortstack{Any \\ Decision}"
label var n_w_dec "\shortstack{No. of \\ Decisions}"
label var sayl "\shortstack{Household \\ Purchases}"
label var hcdec "\shortstack{Health \& \\ Contracept.}"
label var huss "\shortstack{Husband's \\ Money}"
label var saysc "\shortstack{Daily \\ Decisions}"

label var injuries "\shortstack{Any \\ Injury}"
label var n_inj "\shortstack{No. of \\ Injuries}"
label var viol_p2 "\shortstack{Severe \\ Violence}"
label var viol_p1 "\shortstack{Less Severe \\ Violence}"
label var viol_s "\shortstack{Sexual \\ Violence}"
label var viol_e "\shortstack{Emotional \\ Violence}"

label var age_marr "\shortstack{Age at \\ Marriage}"
label var age_gap "\shortstack{Spousal \\ Age Gap}"
label var educ_gap "\shortstack{Spousal \\ Educ. Gap}" 
label var height_diff "\shortstack{Absolute \\ Height Gap}" 
label var workh "\shortstack{Husband \\ Employed}" 
label var h_goodjob "\shortstack{Husband \\ White Collar \\ Job}"   
label var h_earnless "\shortstack{Husband \\Earns Less \\ or Same}" 
label var drunk_often "\shortstack{Husband \\Often Drunk}"

label var educw "\shortstack{Years of \\ Schooling}"
label var primary "\shortstack{Primary \\ School}"
label var height "\shortstack{Height \\(cm)}"
label var lowheight "\shortstack{Low \\ Height}"

label var viol_byothers "\shortstack{Viol. \\ by Others}"
label var viol_natal "\shortstack{Viol. in \\ Natal Family}"
label var told_anyone "\shortstack{Has Told \\ Anyone }"
label var sought_help1 "\shortstack{Help \\ From Family}"
label var sought_help2 "\shortstack{Help From \\ Friends \& \\ Network}"
label var sought_help3 "\shortstack{Help From \\ Police\/Doctor}"

label var scstbc "SC/ST/OBC"
label var rural "Rural"
label var muslim "Muslim"
label var christian "Christian"
label var hindu "Hindu"
label var wealth_index "Wealth Index"
label var hhsize "Household Size"
label var workw "Employed in Past 12 Months"
label var nkids "No. Kids"
label var nkids_home "No. Kids at Home"
label var nomorec "Completed Fertility"
label var kids_home "Kids at Home"


* Separation results

eststo clear
	
	
use "$OUTDATA/NFHS_sample_withdivorce.dta", clear 


cap gen nkids_home = v202 + v203

** Define Separation
cap gen divsep = v501>=4
replace divsep = . if v501 == . | v501 == 3
replace divsep = 1 if husbandelse == 1


cap gen kids_home = nkids_home>0
replace  nomorec = (v614 == v220) 
label var divsep "\shortstack{Divorced/\\ Separated}"
label var husbandelse "\shortstack{Husband \\ Elsewhere}"
label var nkids_home "No. Kids"
label var kids_home "Any Kid"

cap gen firstbornboy = (b4_01 == 1)
replace firstbornboy = . if b4_01 == .
label var firstbornboy "First Born Boy"

cls
foreach y in divsep {

	eststo: reg `y' i.dowTreat dow_cohort hindu1 $cov $fe2 [pweight = v005]  , r cl($cluster )
	qui sum `y' if e(sample)
	estadd scalar m =r(mean)
	
	eststo: reg `y' i.dowTreat##c.nkids_home dow_cohort  hindu1 $cov $fe2 [pweight = v005] , r cl($cluster )
	qui sum `y' if e(sample)
	estadd scalar m =r(mean)
		
		
	eststo: reg `y' i.dowTreat dow_cohort hindu1 $cov $fe2 [pweight = v005] if northeast == 0 & south == 0 & rural == 1, r cl($cluster )
	qui sum `y' if e(sample)
	estadd scalar m =r(mean)
	
	eststo: reg `y' i.dowTreat##c.nkids_home dow_cohort hindu1 $cov $fe2 [pweight = v005] if northeast == 0 & south == 0 & rural == 1, r cl($cluster )
	qui sum `y' if e(sample)
	estadd scalar m =r(mean)
	
	eststo: reg `y' i.dowTreat dow_cohort hindu1 $cov $fe2 [pweight = v005] if northeast == 1| south == 1 | rural == 0, r cl($cluster )
	qui sum `y' if e(sample)
	estadd scalar m =r(mean)
	
	eststo: reg `y' i.dowTreat##c.nkids_home dow_cohort hindu1 $cov $fe2 [pweight = v005] if northeast == 1| south == 1 | rural == 0, r cl($cluster )
	qui sum `y' if e(sample)
	estadd scalar m =r(mean)
	


	}
	

esttab using "$OUTDIR/NFHS3_sep.tex", ///
booktabs nonotes replace indicate("\hline\vspace{-3mm} \\ \vspace{-3mm}  Individual Controls = $cov " " \vspace{-3mm} State FE = *.state " " Year of Birth FE = *.yearb ", labels("Yes" "$-$") ) ///
keep(1.dowTreat 1.dowTreat#c.nkids_home ) ///
label interaction(" \$\times\$ ") substitute("=1" "") ///
stats(N r2 m, labels("Obs." "R sq." "Mean Dep. Var.") fmt(%9.0fc %9.3fc))   ///
se star(* 0.10 ** 0.05 *** 0.01)  b(%9.3fc) se(%9.3fc) 

